
Have you ever had the need to retrieve serial numbers of various components on a computer? How about just trying to figure out what kinds of hardware or software are installed? WMI can help you with these problems and a lot more! This session will give an overview of WMI, including samples of the kinds of information it can provide and how to use it.
About this session

WMI is a pretty big topic. There are lots of new terms to learn and a lot of layers involved. I don’t claim to be an expert with WMI. (Frankly, a lot of this stuff makes my head hurt!) But I think it’s a really important topic to introduce to VFP developers. WMI is very powerful, and has been ignored by most of the VFP community. This session provides a very high-flying overview of the topic. I hope it will inspire you to do further research.
Included code

Most of the code examples included in this session are functions in Demo.prg. Demo.prg also has the ability to list the functions available if you call it like this:

demo("?")
Keep in mind that some of the functions shown are hidden and will generate an error. (Unfortunately, APROCINFO won’t tell me if a method is hidden or not.)

.h Files

The documentation for WMI frequently references .h files, but doesn’t tell you where you can find .h files. It seems that Visual Studio is the only place to find a lot of these files. Knowing that not all of you have a copy of Visual Studio, I’ve placed a file that has all the .h files from Visual Studio on my web site at http://www.peisch.com/downloads/win32_api_includes.zip. The file is over 9 megabytes so I consider it too big to include with the session files. This file was graciously provided to me by Carlos Alloatti of Argentina.
Acronyms

As you’ve probably learned by now, Microsoft is very fond of acronyms, and particularly TLAs (three-letter acronyms). Here are acronyms I’ll be using in this paper. I’ve included links to pages that give you more complete definitions of these.
WMI – Windows Management Instrumentation

CIM – Common Information Model http://msdn.microsoft.com/en-us/library/aa389234(VS.85).aspx
IDL – Interface Definition Language http://msdn.microsoft.com/en-us/library/aa367091(VS.85).aspx 
MOF – Managed Object Format http://msdn.microsoft.com/en-us/library/aa823192(VS.85).aspx
WBEM – Web-Based Enterprise Management (Often seen with “S” in front.)
WQL – WMI Query Language. The syntax of WMI queries http://msdn.microsoft.com/en-us/library/aa394606(VS.85).aspx 
What is WMI?
From the Microsoft Web site:

“Windows Management Instrumentation (WMI) is the infrastructure for management data and operations on Windows-based operating systems. You can write WMI scripts or applications to automate administrative tasks on remote computers but WMI also supplies management data to other parts of the operating system and products, for example System Center Operations Manager, formerly Microsoft Operations Manager (MOM), or Windows Remote Management (WinRM).”

Wikipedia also has a good entry on WMI: http://en.wikipedia.org/wiki/Windows_Management_Instrumentation 

Think of WMI as an API system on steroids. It lets you query just about anything you wanted to know about a computer but it also lets you monitor system status and run scripts. This includes everything from retrieving serial numbers of various components to the getting a list of software that’s installed. It is also capable of performing a lot of tasks, like sending an e-mail whenever a certain event occurs or uninstalling software. WMI continues to be enhanced with each new OS.  It runs on every Windows platform since Windows 98, though not every function in WMI is available on every platform. What you need to do on platforms before Windows 2000 is install the WMI core. Downloads for the core are available from: 

 http://www.microsoft.com/downloads/details.aspx?FamilyId=98A4C5BA-337B-4E92-8C18-A63847760EA5&displaylang=en 

and http://www.microsoft.com/downloads/details.aspx?displaylang=en&FamilyID=AFE41F46-E213-4CBF-9C5B-FBF236E0E875. 
Just to give you a rough idea of what WMI is good for, here’s a list of the kinds of tasks you can perform.

	Task Categories
	Description

	Accounts and Domains
	Obtain information such as the computer domain or the currently logged-on user. Many domain- or account-related tasks are best performed with ADSI scripts. For examples, see the TechNet ScriptCenter at http://www.microsoft.com/technet.

	Computer Hardware
	Obtain information about the presence, state, or properties of hardware components. For example, you can determine whether a computer is a desktop or laptop.

	Computer Software
	Obtain information such as which software is installed by the Windows Installer (MSI) and software versions.

	Connecting to the WMI Service
	To get data from WMI, either on the local computer or from a remote computer, you must connect to the WMI service by connecting to a specific namespace. In most cases, use either the shorthand moniker connection or the Locator connection. 

	Dates and Times
	Windows XP introduced several WMI classes and a scripting object to parse or convert the CIM datetime format.

	Desktop Management
	Obtain data from or control remote desktops. For example, you can determine whether or not the screensaver requires a password. WMI also gives you the ability shut down a remote computer.

	Disks and File Systems
	Obtain information about disk drive hardware state, logical volumes. 

	Event Logs
	Obtain event data from NT Event log files and perform operations like backing up or clearing log files.

	Files and Folders
	Change file or folder properties through WMI, including creating a share or renaming a file.

	Networking
	Manage and obtain information about connections and IP or MAC addresses.

	Operating Systems
	Obtain information about the operating system such as version, whether it is activated, or which hotfixes are installed.

	Performance Monitoring
	Use the WMI classes that obtain data from performance counters to access and refresh data about computer performance.

	Processes
	Obtain information such as the account under which a process is running. You can perform actions like creating processes.

	Printers and Printing
	Manage and obtain data about printers, such as finding or setting the default printer.

	Registry
	Create and modify registry keys and values.

	Scheduled Tasks
	Create and get information about scheduled tasks.

	Services
	Obtain information about services, including dependent or antecedent services.


Namespaces

Namespaces are areas created in Windows for to which you can control access through security settings. I’ll cover how to see what these name spaces are and how to set security later on in the Security section of this session. For now, you just need to be aware that WMI providers and classes exist in specific namespaces.
WMI System Classes

WMI System Classes (or Intrinsic Event Classes) are based on the Common Information Model (CIM). WMI creates a set of these classes for each namespace created. All System Class names start with a double underscore (__) to distinguish them from provider-based classes. System Classes are more complex to deal with so we’ll save them for later. But here’s a reference to some info about them: http://msdn.microsoft.com/en-us/library/aa394583.aspx 
WMI providers

WMI providers are broad categories of tasks based on Managed Object Format (MOF) and provide access to events outside of the WMI system. The classes available through providers are called Extrinsic Event Classes. See http://msdn.microsoft.com/en-us/library/aa394570(VS.85).aspx for more info. The syntax to access these classes is simpler than that for the system classes, and we will focus mainly on these classes in this session.
Most of the time you’ll be dealing with classes based on a provider. WMI providers are categories of tasks available through WMI. Most WMI classes belong to a provider. These types of classes are called Extrinsic Event Classes. Some examples of providers are:

· Active Directory Provider

· Performance Monitoring Provider

· Session Provider (For managing network sessions)

· System Registry Provider

· Win32 Provider

This is just a very small portion of the providers available. For a more complete list of standard providers, see http://msdn.microsoft.com/en-us/library/aa394570.aspx. You may also have providers created by third parties or even some you’ve written yourself.

WMI provider classes

Each of the providers implements one or more classes. These classes are what you reference in your queries and method calls. As an example, the WIN32 provider has an enormous number of classes. Some of these are:
· Win32_BIOS – For reading BIOS settings on your computer
· Win32_ComputerShutdownEvent – To detect when your computer is shutting down and the kind of shutdown requested (User log-off, complete shutdown or restart).

· Win32_ComputerSystem – Has all sorts of information about your computer, including:

· The manufacturer and model
· Whether or not you’ve enabled daylight savings time
· Whether or not the button on the front of your computer lets you reboot
· Whether or not there’s an infrared port on your computer

· Name of the workgroup for this computer (if any)

· Many, many other items too numerous to list here

· Win32_DefragAnalysis – Information about how fragmented your disk drive is

· Win32_Fan – Information about the cooling fan in your computer like when it was installed and what its last error code was and what its current status is

The list goes on and on from there. Each of these classes will return a structure to you that has information, or implements methods you can use. These are all documented on the Microsoft web site.
We don’t necessarily need WMI for all of these things. Some are easier to find out using native functions in VFP. For example, the first item in the table tells us we can use WMI to retrieve the currently logged-on user, but in VFP it’s a lot easier to use GETENV(“USERNAME”) for this.

Starting and stopping WMI

WMI is usually started automatically when your system starts, but you can start and stop it manually. For details on how to do this see http://msdn.microsoft.com/en-us/library/aa826517(VS.85).aspx, but you just use a NET START WINMGMT or NET STOP WINMGMT command.

WMI is always listed as one of the services. See Figure 1. If it’s running, the Status will say, “Started” and will be blank if it not running. You can get to this window by using Control Panel…Administrative Tools…Component Services, then click on the Services node.

Figure 1 – A list of services showing WMI

Because WMI can be stopped, you’ll probably want to confirm it’s running before you try to do much with it. You can do so in code as follows:

TRY 

   * The following line will start the service if it's not running and is 

   * set to manual or automatic.

   * It will fail if the service is disabled.

   loWMI = GETOBJECT('WinMgmts:')

   MESSAGEBOX("WMI is running. You may proceed.")

CATCH

   MESSAGEBOX("WMI isn't running")

ENDTRY 


loWMI = NULL
&& Cleanup after we've done whatever we're going to do

In order to save space, I’m not using the above code in each of my examples in this paper. Please keep this in mind when you copy any of my code for use in your application. The samples on disk all use a variant of the code shown above.
WMI requirements

The requirements for using any particular WMI call vary widely from method to method. The best way to know what you’ll need is to refer to the documentation for the class on http://msdn.microsoft.com/en-us/library/aa394554(VS.85).aspx. Each topic will show you the methods supported and class structure along with possible values of constants used, the operating systems in which it’s available and required DLL and H files. (Note: I’ve found it impossible to find a reliable source for H files other than to install Visual Studio.)

Calling WMI Queries
Use GETOBJECT to get a handle to the Windows WMI service, like so:
loWMI
= GETOBJECT('WinMgmts:')

WMI has a query language which is a subset of the SQL we’re used to, called WQL. We call the ExecQuery method of WMI is how we execute these SQL commands. Instead of pulling data from tables like we do in VFP, we pull data from a “class” instead. This means we need to determine which class has the kind of information we’re looking for. If processor information is what you’re after, the Win32_Processor class will return a collection that holds information for each processor in the current machine.  
loProcessorInfo = loWMI.ExecQuery('Select * from Win32_Processor')

loProcessorInfo is now a reference to a collection of processor information.  This object has a Count property that tells us the number of processors in the collection. We can use the VFP FOR EACH construct to view the information for each processor. The format of the information for each processor in the collection looks something like a table structure. It has field names, which we will use to access specific pieces of information about the processor. The structure of this class looks like this: 

class Win32_Processor : CIM_Processor

{

  uint16 AddressWidth;

  uint16 Architecture;

  uint16 Availability;

  string Caption;

  uint32 ConfigManagerErrorCode;

  boolean ConfigManagerUserConfig;

  uint16 CpuStatus;

  string CreationClassName;

  uint32 CurrentClockSpeed;

  uint16 CurrentVoltage;

  uint16 DataWidth;

  string Description;

  string DeviceID;

  boolean ErrorCleared;

  string ErrorDescription;

  uint32 ExtClock;

  uint16 Family;

  datetime InstallDate;

  uint32 L2CacheSize;

  uint32 L2CacheSpeed;

  uint32 LastErrorCode;

  uint16 Level;

  uint16 LoadPercentage;

  string Manufacturer;

  uint32 MaxClockSpeed;

  string Name;

  string OtherFamilyDescription;

  string PNPDeviceID;

  uint16 PowerManagementCapabilities[];

  boolean PowerManagementSupported;

  string ProcessorId;

  uint16 ProcessorType;

  uint16 Revision;

  string Role;

  string SocketDesignation;

  string Status;

  uint16 StatusInfo;

  string Stepping;

  string SystemCreationClassName;

  string SystemName;

  string UniqueId;

  uint16 UpgradeMethod;

  string Version;

  uint32 VoltageCaps;

};

Now that we have a reference to each processor, and knowing this structure, we can find the ProcessorId for each processor as follows:
FOR EACH loProcessor IN loProcessorInfo

   lcProcessorID = loProcessor.ProcessorId

ENDFOR
If what you’re after is the clock speed instead of the processor ID, you would query the CurrentClockSpeed field in this class instead of the ProcessorID field.
You can also specify to only include the field(s) you are interested in the query. For example:

Another WMI class you might find useful is the Win32_CDROMDrive class. Once again, using ExecQuery to retrieve a collection for CD information:
loWMI
= GETOBJECT('WinMgmts:')

loCDs = loWMI.ExecQuery('Select * from Win32_CDROMDrive')

FOR EACH loDrive IN loCDs

   ? loDrive.Name

   ? loDrive.Drive

ENDFOR 

Want to know if a computer has a floppy drive or not? Just run this code:

loWMI
= GETOBJECT('WinMgmts:')

loFloppies = loWMI.ExecQuery('Select * from Win32_FloppyDrive')

DO CASE 

CASE loFloppies.Count = 0

   MESSAGEBOX("This computer does NOT have a floppy drive.")

CASE loFloppies.Count = 1

   MESSAGEBOX("This computer has a single floppy drive."

OTHERWISE

   MESSAGEBOX("This computer has "+TRANSFORM(loFloppies.Count)+ ;

              " floppy drives.")

ENDCASE

There is a Win32_Product class that could be incredibly useful for checking for installed software. Its use looks something like this:

loWMI
= GETOBJECT('WinMgmts:')

loProducts = loWMI.ExecQuery('Select * from Win32_Product')

FOR EACH loProduct IN loProducts

   IF loProduct.Name = TheNameYou'reLookingFor

      * Do something

   ENDIF

ENDFOR

In addition to the name of the product, this class provides information like the version of the software installed, its location and even the Vendor’s name.

Note that I said this “could be” incredibly useful. The problem is that this code will only work for software installed with the Windows Installer (previously known as MSI). Not only do a lot of software vendors shy away from installing with the Windows installer (many of you probably use Inno Setup for example), but Microsoft itself is getting away from the Windows installer. For anything written in .NET 2.0 or later, Microsoft encourages you to use the ClickOnce technology for installing software.

So, what to use instead? The only solution I’ve found is one that looks through the HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\Windows\CurrentVersion\Uninstall portion of the registry.
http://groups.msn.com/WindowsScript/moffilesfromthemicrosoftscriptingguys.msnw 

Still, if your package does use the Windows Installer (as InstallShield Express does), the Win32_Product class offers methods to install, uninstall and upgrade your software.

Calling WMI Methods
In order to call a method in a WMI class, you must first query the class to retrieve the properties and methods. You can then call the method you want. See http://msdn.microsoft.com/en-us/library/aa384832(VS.85).aspx. Let’s look at an example.
There is a WIN32_Printer class that has a PausePrinter method. So if you want to pause a printer queue, you would do something like this:

FUNCTION PausePrinter

LOCAL loWMI, lcPrinter, loPrinters, loPrinter, lcQuery

loWMI = This.isWMIRunning()

IF NOT ISNULL(loWMI)

   lcPrinter = GETPRINTER()

   IF NOT EMPTY(lcPrinter)

      lcQuery = [Select * from Win32_Printer where Caption = ']+lcPrinter+[']

      loPrinters = loWMI.ExecQuery(lcQuery)

      FOR EACH loPrinter IN loPrinters

         IF loPrinter.Pause() = 0

            MESSAGEBOX(lcPrinter+" has been paused.",64,"Success")

         ELSE

            MESSAGEBOX(lcPrinter+" could not be paused.",48,"Failure")

         ENDIF 

      ENDFOR 

   ENDIF 

ENDIF 

ENDFUNC 

There is a ResumePrinter method to start the printer queue back up again.

Likewise, there is a WIN32_Service class with a variety of methods, including Stop, Start, Create and Delete. So you can use WMI to start and stop Windows Services, or even to add and delete services.

Flags

All of the methods in WMI accept flags as one of the parameters. The parameter position that represents these flags differs from method to method. You must refer to the documentation for each method to determine which parameter accepts the flags. These flags are referenced in all the MSDN documentation as constants which are defined in Wbemdisp.h.

With the ExecQuery method the third parameter is the setting for the flags. 
Documentation for the possible values for flags can be found at http://msdn.microsoft.com/en-us/library/aa393980(VS.85).aspx and http://msdn.microsoft.com/en-us/library/aa393257(VS.85).aspx. The next section talks about how these flags can be useful.

Modes

The examples of using ExecQuery method we’ve used up until this point make semisynchronous calls to WMI. WMI also supports synchronous and asynchronous calls. See http://msdn.microsoft.com/en-us/library/aa384832(VS.85).aspx.The difference between each of these methods is:

Synchronous – Causes the program to wait until results are returned before continuing. Use this method for returning smaller results only.
Asynchronous – Program execution continues while the WMI request is processing. Each object is returned to a SWbemSink object. This is a good mode for monitoring your system in the background, but it is only useful for the local computer.
Semisynchronous – Like asynchronous mode, your program will continue execution while the query is running, but like synchronous mode, the results are returned to the calling program instead of a sink object. (This isn’t entirely accurate in a single-threaded environment like VFP. The process does end up waiting for the Semisynchronous query to finish.) Longer queries can use up your system resources in this mode, so it should be used with the same care as synchronous mode. This mode is more secure than Asynchronous mode, so this is the mode that you should use when running remote queries.
WMI limits which modes can be used with each method. 
If a method supports asynchronous mode, you would call it asynchronously by adding “Async” to the end of the method. For example:
· loWMI.DeleteAsync

· loWMI.ExecQueryAsync

· loWMI.ExecNotificationQueryAsync

Without the “Async” at the end of the method the method will run in Semisynchronous mode by default. In order to run a method in Synchronous mode you must set the wbemFlagReturnWhenComplete flag when you call the method. 
Capturing Windows Events with WMI

This example shows usage of both a System class (Intrinsic Event Class) and an asynchronous WMI call. Because results of an asynchronous call are sent to a sink object, this example also shows how to instantiate and use a sink object.
The first thing you’ll notice is that the syntax for the query of the Win32_PrintJob class in the CatchPrintEvent function uses different syntax. Here we are using the __InstanceCreationEvent System Class, which is documented here: http://msdn.microsoft.com/en-us/library/aa394649.aspx. 

The “within” clause of this query is the polling interval in seconds. Think of this as the Interval property of a timer.

The WHERE clause of this query is specifying that we want to be notified of “events for all classes within the class hierarchy, rather than a specific event class.”

Stuart’s original program just echoed the originating PC and pages to the screen. I added the display of the driver, as well as the QueryPrintEvent method and related code to check on the status of the print job after it has started.

Note that the pages count displayed is the count of pages that have been sent to the spooler. It’s possible for the event to be trapped before the app that sent the print job is finished sending pages to the spooler, and you may see a count less than the actual number of pages.

FUNCTION CatchPrintEvent

* Provided by Stuart Dunkeld and modified by Barbara Peisch

* When a new printjob is started, echo the originating pc and page count

* RELEASE oEventCatcher in the command window to stop.

LOCAL loEventCatcher, lcQuery, llXit

loEventCatcher = createobject("eventcatcher",This)

* __InstanceCreationEvent is a Systems class (Intrinsic Event Class)

lcQuery = [select * from __InstanceCreationEvent within 1] + ;

          [ where TargetInstance ISA 'Win32_PrintJob']

loEventCatcher.CatchEvents(lcQuery)

DECLARE integer Sleep IN kernel32 as Sleep integer

llXit = .F.

SET ESCAPE ON 

ON ESCAPE llXit = .T.

DO WHILE NOT llXit

   IF This.lPrintJobRunning

      This.QueryPrintEvent()

   ENDIF 

   DOEVENTS 

   =sleep(500)

ENDDO 

ON ESCAPE 

RETURN

ENDFUNC 

***************************************************************************

HIDDEN FUNCTION QueryPrintEvent

LOCAL lcStatus

* Note: For simplicity's sake, I'm querying the Win32_PrintJob class again.

* You may prefer to use the Windows GetPrinter() API (not VFP's GETPRINTER() function) to get a more specific printer status. Doing so is rather complex

* because it requires 2 calls to GetPrinter()--one to get the buffer size for

* the PRINT_INFO_n buffer and one to populate it. It may also require a call

* to DocumentProperties() to populate the DevMode structure. If you want to

* look into this technique see http://support.microsoft.com/kb/140285
* for examples in VB.

loWMI = This.isWMIRunning()

IF NOT ISNULL(loWMI)

   loPrintJobs =  loWMI.ExecQuery("Select * from Win32_PrintJob ")

   FOR EACH loPrintJob IN loPrintJobs

      IF NOT INLIST(UPPER(loPrintJob.Status),"OK","UNKNOWN")

         _screen.AlwaysOnTop = .T.

         MESSAGEBOX("There is a problem with your print job." + ;

                     CHR(13)+"Status: "+loPrintJob.Status)

         _screen.AlwaysOnTop = .F.

         This.lPrintJobRunning = .F.

      ENDIF 

   ENDFOR  

ENDIF 

RETURN

ENDFUNC 

*******************************************************************************

** Event catching class

define class eventcatcher as relation

   oSink = .null.

   oWbemSink = .null.

   oWMI = .null.

   oCaller = .null.

   *-----------------------------------------------------------------------

   PROCEDURE Init(toCaller as Object)

   this.oWMI = getobject("winmgmts:")

   * Create the sink objects and bind them

   this.oWbemSink = createobject("wbemscripting.swbemsink")

   this.oSink = createobject("vfpsink",toCaller)

   eventhandler(this.oWbemSink, this.oSink)

   ENDPROC
   *-----------------------------------------------------------------------

   PROCEDURE destroy

   if vartype(this.oWbemSink) = "O"

      this.oWbemSink.Cancel()

   endif

   ENDPROC
   *-----------------------------------------------------------------------

   PROCEDURE CatchEvents

   lparameters cQuery

   * Send the WMI sink not the VFP one.

   * Here is where we use an Asynchronous call

   this.oWMI.ExecNotificationQueryAsync(this.oWbemSink, cQuery)

   ENDPROC
   *-----------------------------------------------------------------------

enddefine

******************************************************************************
** The sink class is easier to manage when decoupled from the eventcatcher class

define class vfpsink as relation

   oCaller = .null.

   implements ISWbemSinkEvents in "WbemScripting.SWbemSink"

   *-----------------------------------------------------------------------

   FUNCTION Init(toCaller as Object)

   This.oCaller = toCaller

   ENDFUNC 

   *-----------------------------------------------------------------------

   PROCEDURE ISWbemSinkEvents_OnObjectReady(oObject, oAsyncContext)

   * A print job has been started

   This.oCaller.lPrintJobRunning = .T.

   This.oCaller.cPrinterName = oObject.TargetInstance.Drivername

   ? "Originating PC: " + oObject.TargetInstance.HostPrintQueue

   ? "Driver: "+oObject.TargetInstance.DriverName

   ? "Pages: " + transform(oObject.TargetInstance.TotalPages)

   ENDPROC
   *-----------------------------------------------------------------------
   * Each of the following PROCEDURE lines needs to be here for the interface.
   PROCEDURE ISWbemSinkEvents_OnCompleted(nResult,oErrorObject,oAsyncContext)

   PROCEDURE ISWbemSinkEvents_OnProgress(nUpperBound, nCurrent, cMessage, oAsyncContext)

   PROCEDURE ISWbemSinkEvents_OnObjectPut(oObjectPath, oAsyncContext)

enddefine
******************************************************************************

For additional information on event queries see: http://msdn.microsoft.com/en-us/library/aa393277(VS.85).aspx 
Security

You can’t do anything with WMI that you wouldn’t have permission to do otherwise. For example, if the user you are logged in as doesn’t have permission to read the registry, you won’t be able to do it with WMI either. You can use WMI to impersonate a different user, but you must know the login name and password. In the “Connecting to a Remote Computer” section we’ll cover how to specify the username and password to use.
The syntax I’ve used until now has been simplified. It assumes you want to run WMI as the currently logged in user on the local machine, with the default impersonation level and default authentication level. Here’s the more verbose way of getting an object reference to WMI:
oWMI = GetObject("WinMgmts:{impersonationLevel=impersonate," + ;

                        "authenticationLevel=pktPrivacy}" + ;

                        "!\\.\root\cimv2")

The “winmngnts:” part should look familiar. It’s the same shortcut, or moniker we’ve been using all along.

Setting impersonationLevel = impersonate is the default. This allows you to use the credentials of the currently logged in user. Other possible settings are Anonymous, Identify and Delegate. These are constants defined in Wbemdisp.h.
Setting authenticationLevel=pckPrivacy is once again the default. Other settings are Default, None, Connect, Call, Pck and PckIntegrity. These are constants defined in Wbemdisp.h.
\\. Is the local computer and \root\cimv2 is the namespace where most WMI classes are defined.
It’s best not to change the default impersonation level or authentication level because of increased security risk, chances of failure and differences between operating systems. If you’re curious as to the details, you can read these here for impersonation:
· http://msdn.microsoft.com/en-us/library/aa393981.aspx 

· http://msdn.microsoft.com/en-us/library/aa393609(VS.85).aspx 

· http://msdn.microsoft.com/en-us/library/aa393617(VS.85).aspx#_hmm_changing_the_default_impersonation_levels_using_c_ 

and here for authentication level:

· http://msdn.microsoft.com/en-us/library/aa393618(VS.85).aspx 

· http://msdn.microsoft.com/en-us/library/aa393851(VS.85).aspx 

· http://msdn.microsoft.com/en-us/library/aa393609(VS.85).aspx 

Setting security for remote access

In order to access a remote computer with WMI, you must be an Administrator on the local machine and have a valid username and password for the remote machine. In addition, there are several things that need to be setup on the remote computer.

These are outlined at http://msdn.microsoft.com/en-us/library/aa393266(VS.85).aspx but I’ll touch on them briefly here as well.

In order to keep from getting too verbose, I’ll only cover XP and higher in this session. If you need information about older operating systems check http://msdn.microsoft.com/en-us/library/aa389284(VS.85).aspx. 

COM/DCOM settings

You can configure DCOM by typing DCOMCNFG into the Windows Start…Run command. The dialog displayed varies with the operating system. With XP and later, DCOM is part of the component services and will display a dialog like that shown in Figure 1.
Expand the Component Services node and then right-click on My Computer and choose Properties. You will see a dialog like that shown in Figure 2.

Figure 2 – Setting DCOM in XP and later
Click on the COM Security tab. Click on the Edit Limits… buttons in each section and allow remote access, and remote launch and activation for the user you want to have access to this machine.
Allow access to WMI namespaces

The quickest way to access the dialog for configuring WMI properties is to type WMIMGMT.MSC into the Start…Run command. A dialog like that shown in Figure 3 is shown. 

Figure 3 – WMI configuration dialog
Right-click on the WMI Control (Local) icon and select Properties. Then select the Security tab. This shows all the namespaces available. (See Figure 4.) Click on the Security button to set the access rights for individual users.

Figure 4 – Namespaces available to WMI

Connecting to a remote computer

Connecting to a remote computer is probably the trickiest thing you can do with WMI. That’s because there are so many variables based on things like which operating system (OS) the remote computer is running and whether or not it’s running a firewall. In Vista you have to worry about settings for the User Account Control (UAC). 
As I mentioned in the Security section of this paper, in order to access a remote computer with WMI, you must be an Administrator on the local machine and have a valid username and password for the remote machine.

In the Security section of this paper I showed you how to specify the impersonation level and authentication level, but you may have noticed that there was nowhere in this syntax to specify a username and password. In order to do that, we need to call WMI using the older syntax we needed to use before VFP 7:

oWMILocator = createobject("WbemScripting.SWbemLocator")

oWMIServices = oWMILocator.ConnectServer ;



(ComputerName, NameSpace, UserName, Password, ;



 Locale, Authority, Flags)

The good news is, from this point on, there’s really no difference in how you call WMI queries or methods.
 Here are some references to help you research this topic:
http://msdn.microsoft.com/en-us/library/aa384463(VS.85).aspx 

http://msdn.microsoft.com/en-us/library/aa389290.aspx 

Here is an example taken from Stuart Dunkeld’s blog at http://weblogs.foxite.com/stuartdunkeld/archive/2005/09/14/910.aspx.  
If you want to connect as the currently logged in user, leave the username and password blank. Note that depending on the type of authentication the network is configured to use, the password may be sent over the network in plain text, so it’s best not to send the password unless absolutely necessary. 
If you want to connect to the local machine, but with a different username and password from the currently logged in user, specify the username and password, but leave the computer name blank. Stuart’s code will substitute the name with ‘.’ in this case. 
Stuart does a good job of explaining what’s going on with comments in the code, so I won’t elaborate on this any further.
** Example usage: query a remote PC for the startup commands buried in the registry

oWMIConnection = createobject("WMIConnect")

oWMIConnection.cDomain = "domain"

oWMIConnection.cUsername = "remoteadmin"

* Computer can be specified by IP address or DNS name.

oWMIConnection.cComputer = "target" 

oWMIConnection.cPassword = "itsasecret"

oWMI = oWMIConnection.GetConnection()

if vartype(oWMI) = "O"

   oCommands = oWMI.ExecQuery("Select * from WIN32_StartupCommand")

   ? transform(oCommands.Count) + " startup commands found"

endif

** WMI connection class

define class WMIConnect as relation

   ** User/domain settings

   cDomain = ""

   cUserName = ""

   cPassword = ""

   ** Target computer. 

   cComputer = "" 

   ** WMI settings

   cNameSpace = "root/cimv2"

   * Without this flag, a connection attempt would wait indefinitely.

   * nFlags = 128 && XP or better only..

   nFlags = 0

   * Find your Locale ID here: 

   * http://www.microsoft.com/globaldev/reference/win2k/setup/lcid.mspx
   cLocale = "MS_409" && EN-US

   ** Security settings.

   nImpersonationLevel = 3 && wbemImpersonationLevelImpersonate

   nAuthenticationLevel = 0 && wbemAuthenticationLevelDefault

   * If .T., load all privileges before connection.

   lLoadPrivileges = .f.

   procedure GetConnection

      local cUserName, cPassword, lRemote, oWMILocator, ;

            oPrivileges, oWMIServices, nPrivilegeID

      if empty(this.cComputer)

         * Use "." for a local connection

         this.cComputer = "."

      endif

      if this.cComputer = "."

         * Cannot use credentials to connect locally

         * even if they are the ones currently in use.

         cUserName = ""

         cPassword = ""

         lRemote = .f.

      else

         cUserName = iif(not empty(this.cDomain), this.cDomain + "\", "") ;

                        + this.cUserName

         cPassword = this.cPassword

         lRemote = .t.

      endif

      oWMILocator = createobject("WbemScripting.SWbemLocator")

      if lRemote

         * Apply settings for DCOM:

         oWMILocator.Security_.ImpersonationLevel = this.nImpersonationLevel

         oWMILocator.Security_.AuthenticationLevel = this.nAuthenticationLevel

      else

         * Privileges only need to be applied for on local system

         * but must be requested *before* connection on Windows 9x/NT

         if this.lLoadPrivileges

            for nPrivilegeID = 1 to 27

               oWMILocator.Security_.Privileges.add(nPrivilegeID)

            next

         endif

      endif

      cAuthority = "" && for kerberos, see PSDK

      * Connect as specified...

      oWMIServices = oWMILocator.ConnectServer ;

                    (this.cComputer, this.cNameSpace, cUserName, cPassword, ;

                     this.cLocale, cAuthority, this.nFlags)

      * ... and return the WMI service object.

      return oWMIServices

   endproc

   procedure error(nError, cMethod, nLine)

      * Sample error code

      * This needs to be tested across versions to make sure 

      * same errors are raised.

      local arrErr[1]

      aerror(arrErr)

      display memory like arrErr

      if nError = 1429 or nError = 1427

         nError = arrErr(7)

      else

         messagebox("Error " + transform(nError) + " has occurred")

         * Oops..

      endif

      do case



      case nError = 4110

         ? "Invalid namespace"

      case nError = 4099

         ? "Valid credentials, but not allowed to connect"

      case nError = 5

         ? "Invalid credentials"

      case nError = 1722

         ? "Remote computer not found" 

      endcase

   endproc

enddefine

WMI Administrative Tools

There are some tools available for WMI that might help you with the learning curve. Your first impulse might be to find an SDK for WMI, but I can tell you that if you do a search in MSDN for WMI SDK, it will bring you to a download page that is VERY old. (It pre-dates Windows 2000!)

What you want is to download the set of WMI Administrative Tools. You can find that here:
http://www.microsoft.com/downloads/details.aspx?FamilyID=6430f853-1120-48db-8cc5-f2abdc3ed314&DisplayLang=en  

What this will give you is the following tools:

· WMI CIM Studio (This includes a more up-to-date SDK)

· WMI Event Registration

· WMI Event Viewer

· WMI Object Browser

When you start up any of these tools, they will ask you for a namespace. This tells WMI which computer you want to connect to and which namespace. See figure 5.

Figure 5 – Namespace for WMI tools
The default shown will connect to your local computer. The button to the right of the dropdown allows you to browse for other computers on your network.

Once you select a computer, you are presented with a login dialog. See figure 6. When connecting to another computer, you may login as a different user. When connecting to your local machine, you can only login as the current user—the checkbox is checked and disabled.

Figure 6 – Login dialog

Unfortunately, I haven’t had much time to learn what these tools can do, so I’ll have to leave that for you to explore.
WMI info on the Web

Microsoft Knowledgebase
The Microsoft knowledgebase has a vast collection of information about WMI. The main page for this is http://msdn2.microsoft.com/en-us/library/aa394582.aspx so I suggest you start there.

WMI for ODBC

You can use WMI in conjunction with ODBC queries. See http://msdn.microsoft.com/en-us/library/aa393647(VS.85).aspx. 
Building your own WMI Providers

If you’re really ambitious and what to try your hand at writing your own providers, I suggest you start here: http://msdn.microsoft.com/en-us/library/cc512238(VS.85).aspx and here: http://msdn.microsoft.com/en-us/library/aa389276(VS.85).aspx 
Stuart Dunkeld’s BLOG

Stuart Dunkeld has done quite a bit with WMI and has blogged about it. Here is a page that references each of the blog entries: http://weblogs.foxite.com/stuartdunkeld/archive/category/1007.aspx 

The first link (“Securing a Remote WMI Connection) on Part 3 of Stuarts blog is broken. The correct link is http://msdn.microsoft.com/en-us/library/aa393266(VS.85).aspx. 
WMI Glossary

http://msdn.microsoft.com/en-us/library/aa394561(VS.85).aspx
Summary

As you can see, WMI is a very expansive topic. My goal with this session has been to give you some idea of the possibilities of WMI, along with many launching points to find more information on many of these possibilities. Happy hunting!
� As of the writing of this document, this was located on � HYPERLINK "http://msdn.microsoft.com/en-us/library/aa394582.aspx" �http://msdn.microsoft.com/en-us/library/aa394582.aspx�.  Be warned that Microsoft pages change often.


� This table was taken from � HYPERLINK "http://msdn2.microsoft.com/en-us/library/Aa394585.aspx" �http://msdn2.microsoft.com/en-us/library/Aa394585.aspx�. 


� This syntax can only be used with VFP 7 or later. If you are running an older version of VFP you must use this syntax instead:


loWMILocator = createobject("WbemScripting.SWbemLocator")


loWMI = loWMILocator.ConnectServer(".", "root\cimv2")





� This quote is taken from � HYPERLINK "http://msdn.microsoft.com/en-us/library/aa391408(VS.85).aspx" �http://msdn.microsoft.com/en-us/library/aa391408(VS.85).aspx� 





